

# Testing Seaside Components

C. David Shaffer  
Department of Mathematics and Computer Science  
Westminster College




# *Features*

- Tests run on server = access to component being tested
  - Uses Smalltalk debugger
  - Web test runner
  - Available for Squeak and VisualWorks (thanks to Michel Bany for VW port!)
- 
-

# *First example*

## Class SCTestComponent1

```
renderContentOn: html  
  html cssId: 'main'.  
  html span: 'hello'
```



# *First example*

```
SCComponentTestCase subclass: #SCSampleComponentTest
instanceVariableNames: ''
classVariableNames: ''
poolDictionaries: ''
category: 'SeasideTesting-Examples'
```

---

---

# *First example*

```
testComponent1
  self newApplicationWithRootClass: SCTestComponent1.
  self establishSession.
  self assert: (self lastResponse
    stringWithId: 'main') = 'hello'
```

---

---

***Following anchors***



# Class SCAnchorDemo


```
renderContentOn: html
  html cssId: 'first'.
  html
    anchorWithAction: [self firstPressed]
    text: 'first link'.
```

```
firstPressed
  self inform: 'pressed'
```

---

---

# SCAnchorDemo





# Following anchors

```
testAnchor
```

```
self newApplicationWithRootClass: SCAnchorDemo.  
self establishSession.
```

```
self followAnchor: (self lastResponse  
    anchorWithId: 'first').
```

```
self assert: (self lastResponse  
    containsString: 'pressed')
```

“alternatively”

```
self assert: (self lastResponse  
    elementsNamed: 'h3') first  
    contentString = 'pressed'
```



# *lastResponse*

Response parsed --> XML DOM (XMLElement)

Wrapped in SCSeasideResponse:

- Convenience methods for searching for XML elements by id, class or name (tag)
- Method for wrapping parts in subclasses of SCXMLElementWrapper. For example:
  - SCSubmitButtonHtmlInput
  - SCTextAreaHtmlInput
  - SCSeasideForm
  - SCSeasideAnchor



# *Finding anchors*

Selected methods of SCSeasideResponse which return SCSeasideAnchor(s)

anchorWithId: -- uses CSS id

anchorWithLabel: -- text inside A tag

anchors – collection of anchors in order of occurrence



# Web TestRunner

Home Bookmarks Smalltalk Java Web Specs Oracle My Sites Linux Erie Plating Shopping Squ

Test runner

1 run, 1 passes, 0 expected failures, 0 failures, 0 errors, 0 unexpected passes

---

**Failed:**

---

**Errors:**

---

**Passed:**

SCAnchorDemoTest>>#testAnchor [Browse History](#)

---

[close](#)

# Web TestRunner



# *Forms*




# Class SCFormDemo

```
renderContentOn: html
  html form: [
    html text: 'Name: ';
    cssId: 'name';
    textInputOn: #name of: self; br;
    text: 'Age: ';
    cssId: 'age';
    textInputOn: #age of: self; br;
    submitButtonWithAction:
      [self displayInfo]text: 'Display' ]
```


---

---

# Class SCFormDemo



A screenshot of a web browser window. The address bar shows a URL starting with 'http://'. Below the address bar, there are navigation buttons (back, forward, refresh, stop) and a bookmark bar with 'Home', 'Bookmarks', and 'Smalltalk'. The main content area contains a form with two input fields: 'Name: Bob Smith' and 'Age: 40'. Below the input fields is a button labeled 'Display'.





# Forms

```
testDisplay
  | form |
  self newApplicationWithRootClass: SCFormDemo.
  self establishSession.
  form := self lastResponse forms first.
  form textInputWithId: 'name' value: 'Bob Smith'.
  form textInputWithId: 'age' value: '40'.
  self
    submitForm: form
    pressingButton: form buttons first.

self assert: (self lastResponse
  elementsNamed: 'h3') first
  contentString =
  'Hello Bob Smith.  You are 40 years old.'
```

---

---

# *What do we test?*

Seaside component more than just visual display

- State: often tests of state are less brittle than tests of displayed content
  - “answer”: components that provide a Seaside answer
  - Callbacks: components that provide hooks
- 
-

Need access to Seaside component



# *Testing the Counter (WACounter)*


#component answers the instance of the component used to satisfy the **last** request

Seaside may be keeping track of several “versions” of that component registered with #registerObjectForBacktracking:

---

---

# Testing the Counter (WACounter)



# Testing the Counter (WACounter)

The screenshot shows the 'System Browser: WACounter' window. The left sidebar lists various packages, with 'Seaside-Examples-Test' selected. The main pane shows a list of classes, with 'WACounter' highlighted. Below the list are buttons for 'instance', '?', and 'class'. The right pane shows the class's methods: 'count', 'decrease', 'increase', 'initialize', and 'renderContentOn:'. Below the browser are buttons for 'browse', 'senders', 'implementors', 'versions', 'inheritance', 'hierarchy', 'inst vars', 'class vars', and 'source'. The main content area displays the following information:

```
WAComponent subclass: #WACounter
  instanceVariableNames: 'count'
  classVariableNames: ''
  poolDictionaries: ''
  category: 'Seaside-Examples-Test'
```

THIS CLASS HAS NO COMMENT!

# *Testing the Counter (WACounter)*

```
testBack
  self newApplicationWithRootClass: WACounter.
  self establishSession.
  self followAnchor: (self lastResponse
    anchorWithLabel: '++').
  self followAnchor: (self lastResponse
    anchorWithLabel: '++').
```

---

---

# *Testing the Counter (WACounter)*

```
testBack
  self newApplicationWithRootClass: WACounter.
  self establishSession.
  self followAnchor: (self lastResponse
    anchorWithLabel: '++').
  self followAnchor: (self lastResponse
    anchorWithLabel: '++').
  self assert: self component count = 2.
```

---

---



# *Testing the Counter (WACounter)*

```
testBack
  self newApplicationWithRootClass: WACounter.
  self establishSession.
  self followAnchor: (self lastResponse
    anchorWithLabel: '++').
  self followAnchor: (self lastResponse
    anchorWithLabel: '++').
  self assert: self component count = 2.
  self back.
```

---

---

# Testing the Counter (WACounter)

```
testBack
  self newApplicationWithRootClass: WACounter.
  self establishSession.
  self followAnchor: (self lastResponse
    anchorWithLabel: '++').
  self followAnchor: (self lastResponse
    anchorWithLabel: '++').
  self assert: self component count = 2.
  self back.
  self
    followAnchor: (self lastResponse
      anchorWithLabel: '++').
  self assert: self component count = 2
```



---

---

# ***Detecting answers***



# Detecting answers



# *Detecting answers*

```
testYes
  | form |
  self newApplicationWithRootClass: WAYesOrNoDialog.
  form := self establishSession forms first.
  self
    submitForm: form
    pressingButton: (form buttonWithValue: 'Yes').
  self assert: self answer
```



# *Detecting answers*

Related methods:

answer – components last answer (error if none)

componentAnswered – boolean, has the component answered?

componentAnswered: value – did the component answer the specified value?



# *Testing callbacks*



# WAMiniCalendar



Must supply canSelectBlock to instance – server creates instance!

Optional selectBlock callback



# *WAMiniCalendar*

```
testSelectedDate
  | selected anchors |
self
  newApplicationWithRootClass: WAMiniCalendar
  initializeWith: [:cal |
    cal canSelectBlock: [:date | true].
    cal selectBlock: [:date | selected := date]].
```

---

---

# WAMiniCalendar

```
testSelectedDate
| selected anchors |
self
  newApplicationWithRootClass: WAMiniCalendar
  initializeWith: [:cal |
    cal canSelectBlock: [:date | true].
    cal selectBlock: [:date | selected := date]].
self establishSession.
self assert: selected isNil.
anchors := self lastResponse
  anchorsWithLabel:
    (Date today dayOfMonth printString).
self followAnchor: anchors last.
self assert: selected = Date today.
```

---

---

## *Other topics...*

- Session also available
- Hook for configuring application
- History in Web TestRunner is “live”
- Marking interactions for visual inspection
- <http://www.cdshaffer.com/david/Seaside>



# *Issues*

- Visual appearance not tested
  - Support for storing snapshots of pages for human testers to view
- Client scripting (Javascript/DHTML) not tested – Squelenium Demo?



# *Other free frameworks*

- SmallHttpUnit
    - VW, runs “outside” server
    - Excellent API for accessing page elements
  - HttpUnit
    - Java, runs “outside” server
  - Cactus
    - Java, designed to run in-container like SeasideTesting
  - StrutsTestCase – like Cactus+HttpUnit
- 
-

# *Conclusions*

- Test components in isolation or in larger application
- Can interact directly with component/session to test state
- Can test back button behavior

